# Copyright (C) 2010 Teppo Eeva
# This file is part of Arithmetic Game
# For more information, see <http://code.google.com/p/arithmetic-game/>
# Arithmetic Game is free software: you can redistribute it and/or modify
# it under the terms of the GNU General Public License as published by
# the Free Software Foundation, either version 3 of the License, or
# (at your option) any later version.
# Arithmetic Game is distributed in the hope that it will be useful,
# but WITHOUT ANY WARRANTY; without even the implied warranty of
# MERCHANTABILITY or FITNESS FOR A PARTICULAR PURPOSE. See the
# GNU General Public License for more details.
# You should have received a copy of the GNU General Public License
# along with Arithmetic Game. If not, see <http://www.gnu.org/licenses/>.
android{DEFINES += Q_ARITHMETIC_GAME_ANDROID}
TARGET = Arithgame
TEMPLATE = app
!android{
DESTDIR = bin
}
SOURCES += mainwindow.cpp \
    main.cpp \
    arithengine.cpp \
    arithquestion.cpp \
    highscoremanager.cpp \
    aboutdialog.cpp \
    settingsdialog.cpp \
    highscoreitem.cpp \
    highscoredialog.cpp
!android{
    SOURCES += audiomanager.cpp
}
HEADERS += mainwindow.h \
    arithengine.h \
    arithquestion.h \
    highscoremanager.h \
    aboutdialog.h \
    settingsdialog.h \
    highscoreitem.h \
    highscoredialog.h \
    arithgame.h
!android{
    HEADERS += audiomanager.h
}

FORMS += aboutdialog.ui \
    settingsdialog.ui \
    highscoredialog.ui

symbian||android {
    # The separate UI for mobile clients
    FORMS += mainwindow_mobile.ui
    
    # Additional sources and headers
    SOURCES += answeredit.cpp
    HEADERS += answeredit.h
    symbian {
    # Deploy the sound files
    soundFiles.sources = bin/button-12.wav \
    bin/button-19.wav \
    bin/bonus.wav \
    bin/gameover.wav
    soundFiles.path = !:\private\a89fa339 # IMPORTANT! Must be changed if the UID changes!
    DEPLOYMENT += soundFiles
    
    # Deploy the language files
    langFiles.sources = bin/arithgame_en.qm \
    bin/arithgame_fi.qm
    langFiles.path = !:\private\a89fa339 # IMPORTANT! Must be changed if the UID changes!
    DEPLOYMENT += langFiles
    
    # Application UID and capabilities
    TARGET.UID3 = 0xA89FA339
    TARGET.CAPABILITY = "None"
    }
} else {
    FORMS += mainwindow.ui
    QT += svg
}

RESOURCES += arithgame.qrc
TRANSLATIONS = arithgame_en.ts \
    arithgame_fi.ts

OTHER_FILES += \
    android/src/eu/licentia/necessitas/industrius/QtApplication.java \
    android/src/eu/licentia/necessitas/industrius/QtActivity.java \
    android/src/eu/licentia/necessitas/industrius/QtSurface.java \
    android/src/eu/licentia/necessitas/ministro/IMinistroCallback.aidl \
    android/src/eu/licentia/necessitas/ministro/IMinistro.aidl \
    android/AndroidManifest.xml \
    android/res/drawable-ldpi/icon.png \
    android/res/values/libs.xml \
    android/res/values/strings.xml \
    android/res/drawable-mdpi/icon.png \
    android/res/drawable-hdpi/icon.png
